

Dowling14.wmx: Integrals & Economics

TABLE OF CONTENTS

PREFACE	2
REFERENCES	3
INVESTMENT AND CAPITAL FORMATION	3
COST FUNCTIONS	8
TOTAL AND MARGINAL REVENUE	15
CONSUMPTION FUNCTION, MARGINAL PROPENSITY TO CONSUME ---	22
SAVINGS FUNCTION AND MARGINAL PROPENSITY TO SAVE	

```
(%i5) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2],
    background_color = light_gray, draw_realpart=false)$
    fprintfprec:5$ ratprint:false$ kill(all)$
```

```
(%i1) load ("Econ1.mac");
```

```
(%o1) c:/work5/Econ1.mac
```

Sample use of functions in Econ1.mac

```
(%i5) z : 3*x^2 - x*y + 2*y^2 - 4*x - 7*y + 12;
critPt : [ x =1,y =2];
Analyze (z, critPt)$
plotCP (z, critPt);
```

```
(z) 2 y^2 - x y - 7 y + 3 x^2 - 4 x + 12
```

```
(critPt) [x=1,y=2]
```


```
1 cp = [x=1,y=2] [relative minimum,value = 3.0]
```

```
secondDeriv = [6,4,-1]
```

```
surface of 2 y^2 - x y - 7 y + 3 x^2 - 4 x + 12
```

```
near critical point = [x=1,y=2]
```

```
(%t5)
```



```
(%o5)
```

1 Preface

This worksheet is one of a number of wxMaxima files available in the section
Economic Analysis with Maxima
on my CSULB webpage.

We use Maxima to solve a few of the problems (and draw some of the plots) in Ch. 14 of the supplemental text: Introduction to Mathematical Economics, 3rd ed, (Schaum's Outline Series), by Edward T. Dowling (1992, 2001), McGraw-Hill. This modestly priced text is a bargain with many worked out examples. You should compare the examples worked out "by hand" in this text with what we do using Maxima. Section numbers [x.y] refer to sections in Dowling's text.

We have slightly changed some of the symbols used in particular problems.

Ted Woollett
<https://web.csulb.edu/~woollett/>
Aug. 27, 2021

2 **References**

D. Wade Hands, Introductory Mathematical Economics, 2nd ed, Oxford Univ. Press, 2004, Ch. 1.

Chiang & Wainwright, Fundamental Methods of Mathematical Economics, 4th ed., 2005, Ch. 14.

3 **Investment & Capital Formation [Dowling 14.6]**

$K(t)$ is the "time path of capital", the amount of capital accumulated at time t .
The first derivative, $dK(t)/dt$, is the "rate of capital formation" = "rate of net investment flow $I(t)$ at time t ".

$$I(t) = dK(t)/dt$$

More briefly, $I(t)$ = "net investment" (at time t). The units of $I(t)$ will then be the units used for capital K divided by the units used for time.

If we let $[x]$ = units of x , and we assume $[K]$ = thousands of \$ and $[t]$ = years, then
 $[I] = [K]/[t]$ = thousands of \$/year.

If we start with the time t' relation

$$I(t') = dK(t')/dt',$$

multiply both sides by dt' and sum (integrate) over values of t' from $t' = t_0$ to $t' = t > t_0$,

we get $\int_{t_0}^t I(t') dt' = \int_{t_0}^t (dK(t')/dt') dt' = K(t) - K(t_0)$,

since $\int dt' I(t') = \int dt' (dK(t')/dt') = \int dK(t') = K(t_f) - K(t_i)$, using t_f for "t-final" and t_i for "t-initial".

Hence,

$$K(t) = K(t_0) + \int_{t_0}^t I(t') dt' = K(t_0) + \int_{t_0}^t I(t') dt'$$

3.1 Ex. 3, Chiang/Wainwright, p. 466

Given the net investment flow at time t :

$$I(t) = 3 t^{1/2} = 3 \sqrt{t}$$

let $t_0 = 0$, $K_0 = K(t_0) = K(0) =$ capital at time $t = 0$.

Then $\Delta K = K(t) - K_0 = \int_{t_0}^t I(t') dt'$ is a "definite integral".

Calculate the increase of capital K over the time interval $[0, t]$.

Here is an example of an "indefinite integral", since we don't include the limits on t ; the result of an indefinite integral is an "antiderivative", differentiating the result once produces the original function.

```
(%i6) integrate (sqrt(t), t);
```

```
(%o6) 2 t3/2
      3
```

```
(%i7) diff (% , t);
```

```
(%o7) sqrt(t)
```

```
(%i8) grind(%)$
```

```
sqrt(t)$
```

In this Ex. 3, we need to calculate a "definite integral", in which we supply the limits of integration. We let $tp = t'$ be the dummy variable of integration here.

We define a Maxima function $I(t)$.

```
(%i11) assume (t > 0);
```

```
I(tt) := 3*sqrt(tt);
```

```
ΔK : integrate (I(tp),tp,0,t);
```

```
(%o9) [t > 0]
```

```
(%o10) I(tt) := 3*sqrt(tt)
```

```
(ΔK) 2 t3/2
```

(%i12) ΔK ;

(%o12) $2 t^{3/2}$

If we had not included the assume ($t > 0$) line above, Maxima would have asked for the sign of the final value of tp , t . We could have then entered p ; and the same answer would have been produced.

3.2 Ex. 4, Chiang/Wainwright, p. 467

If net investment $I(t) = 1000$ \$/yr, what will be the "total net investment" (capital formation) during one year, say from $t = 0$ to $t = 1$ yr?

(%i13) ΔK : integrate (1000, tp, 0, 1);

(ΔK) 1000

In this problem, the units used for time t are years. The increase in capital K in one year is thus \$1000.

3.3 Ex. 5, Chiang/Wainwright, p. 467

Using the rate of capital formation $I(t)$ defined in Ex. 3,

$I(t) = 3 \sqrt{t}$, with the units thousands of \$/yr,

what is the increase in capital ΔK over the time interval $[1,4]$, that is over the second, third, and fourth years?

(%i14) ΔK : integrate (I(tp),tp,1,4);

(ΔK) 14

Thus K has increase by 14 thousand dollars over the time interval $[1 \text{ yr}, 4 \text{ yr}]$

3.4 Dowling Ex. 9, p 332

The rate of net investment is given by

$I(t) = 140 t^{(3/4)}$,

and the initial stock of capital at $t = 0$ is 150. Determine $K(t)$, the time path of capital.

(%i16) assume ($t > 0$);

K : 150 + integrate (140*tp^(3/4), tp, 0, t);

(%o15) [redundant]

(K) $80 t^{7/4} + 150$

Maxima already remembered that we are assuming $t > 0$, so returned [redundant].

What is then the stock of capital at $t = 5$?

```
(%i17) at(K, t = 5);
```

```
(%o17) 16 511/4 + 150
```

```
(%i18) float(%);
```

```
(%o18) 1487.5
```

You get a numerical answer in one line this way (for example):

```
(%i19) at(K, t = 10),numer;
```

```
(%o19) 4648.7
```

or this way:

```
(%i20) float ( at(K, t = 10) );
```

```
(%o20) 4648.7
```

If you want to turn the expression K into a Maxima function, say $Kv(t)$, use two single quotes (') in front of K , as follows:

```
(%i21) Kv(t) := 'K;
```

```
(%o21) Kv(t) := 80 t7/4 + 150
```

```
(%i22) Kv(10),numer;
```

```
(%o22) 4648.7
```

```
(%i23) Kv(12),numer;
```

```
(%o23) 6339.5
```

3.5 Dowling Prob. 14.26

The rate of net investment $I = 60 t^{1/3}$, and the capital stock at $t = 1$ is 85. Find an expression for $K(t)$ with $t \geq 1$.

```
(%i3) killAB()$
      assume (t > 0);
      I(tt) := 60*tt^(1/3);
      K : 85 + integrate ( I(tp), tp, 1, t);
```

```
(%o1) [t>0]
```

```
(%o2) I(tt) := 60 tt1/3
```

```
(K) 60  $\left( \frac{3 t^{4/3}}{4} - \frac{3}{4} \right) + 85$ 
```

```
(%i4) K : ratsimp (K);
```

```
(K) 45 t4/3 + 40
```

```
(%i5) at (K, t = 1);
```

```
(%o5) 85
```

Thus for $t > 1$ we can use $40 + 45 t^{4/3}$ for K .

```
(%i6) float ( [at (K, t = 6), at (K,t = 10) ] );
```

```
(%o6) [530.62, 1009.5]
```

Here is the slope of curve $K(t)$ at time $t = 6$:


```
(%i7) I(6), numer;
```

```
(%o7) 109.03
```

Here is a plot of $K(t)$ with a tangent line drawn (slope $I(6)$) to the curve $K(t)$ at time $t = 6$. Our plot makes use of the `grid` and `background_color` settings defined at the top of this worksheet.

```
(%i8) wxdraw2d (xlabel = "t", ylabel = "K(t)", key_pos = top_left, key = "K(t)",
  explicit (K, t, 1, 10), color = red, key = "tangent at t = 6",
  explicit (1(6)*(t - 6) + subst (6,t,K), t, 3, 9), color = black, key = "",
  point_type = filled_circle, points ([6], [subst(6,t,K)]) )$
```

(%t8)



4 Cost Functions TC, MC, VC, FC, ATC, AVC [Hands, Ch. 1]

4.1 Short Run & Long Run Cost Functions

We roughly follow Hands, Ch. 1, p. 29 ff.

A firm's total-cost function TC defines the total cost of production of various levels of output (y or Q). We can divide the total-cost function into variable costs VC which depend on level of output and fixed costs FC (independent of output).

$$TC(y) = VC(y) + FC.$$

"Short-run costs" involve both fixed and variable costs, while all costs are variable in the long run. As production y increases, $VC(y)$ increases to the point at which we can ignore the relatively smaller fixed cost FC.

Consider a case in which $FC = 100$ and $VC(y) = 4y^2$. Under these conditions, the total-cost function is

$$TC(y) = 4y^2 + 100$$

```
(%i2) killAB()$
TC : 4*y^2 + 100;
[at (TC, y = 1), at (TC, y = 5)];
```

(TC) $4y^2 + 100$

(%o2) [104, 200]

The total cost to produce $y = 1$ is \$104 and the total cost to produce $y = 5$ is \$200, etc.

In addition to total-costs (TC), economists pay attention to average and marginal costs.

4.2 Average Cost Functions ATC, AVC, AFC

In the short run there are three average cost functions: average total cost (ATC), average variable cost (AVC), and average fixed cost (AFC), defined by

$$ATC = TC(y)/y, \quad AVC = VC(y)/y, \quad AFC = FC/y$$

Note that since $TC = VC + FC$, $ATC = AVC + AFC$

For our example, $TC = 4y^2 + 100$,

$$ATC = 4y + 100/y, \quad AVC = 4y, \quad AFC = 100/y.$$

Because FC does not depend on y , AFC gets arbitrarily large as $y \rightarrow 0$, and that limit is not important in Economics.

4.3 Marginal Cost Function MC

If we imagine that we can consider a limit in which the change in output $\Delta y \rightarrow 0$ in an approximate description of economic reality, we can define the "marginal cost function" (MC) as the first derivative of the total cost function with respect to output y .

$MC = d[TC(y)]/dy = \text{diff}(TC, y)$ in Maxima syntax.

Since $TC = VC + FC$, and since $d(FC)/dy = 0$, we have $MC = d(VC)/dy$.

In our example

```
(%i4) VC : 4*y^2;
      MC : diff (VC, y);
```

```
(VC) 4 y^2
```

```
(MC) 8 y
```

We get $MC = 8y$. For this example, both MC ($8y$) and AVC ($4y$) are linear functions of y . Standard introductory economics texts usually display a U shape curve for MC and AVC, which would only show up if TC was a cubic function of production y , such as

$$TC(y) = ay^3 + by^2 + cy + d.$$

4.4 Relating MC to ATC

Since $ATC = TC/y$, rearranging we get $TC = y*ATC$, so $MC = dTC/dy = ATC + y*d(ATC)/dy$. Solving for $d(ATC)/dy$, we get,

$$d(ATC)/dy = (MC - ATC)/y$$

The slope of the average total cost curve ATC is the left hand side of this equation. If MC lies above ATC, $MC > ATC$, then the slope of ATC is positive and ATC is rising. If MC lies below ATC, $MC < ATC$, then the slope of ATC is negative and ATC is declining. If $MC = ATC$, then the slope of ATC is equal to zero.

When the average total cost (ATC) is rising then the marginal cost (MC) must lie above the total average cost ($MC > ATC$).

When the average total cost (ATC) is falling then the marginal cost (MC) must lie below the total average cost ($MC < ATC$).

When the average total cost is at a critical point (zero slope) then $MC = ATC$.

An identical relation exists between marginal cost (MC) and average variable cost (AVC).

4.5 Finding Total Cost from Marginal Cost

This is Example 1 from Chiang & Wainwright, p. 465.

If the marginal cost (MC) of a firm is the following function of output,

$$dC/dQ = 2*exp(0.2*Q)$$

and if the fixed cost $C_f = 90$, find the total cost function $C(Q)$ for any value of $Q \geq 0$.

By integrating the marginal cost (MC) over $Q = 0$ to some arbitrary positive value of Q , and adding the fixed cost C_f , we get the total cost function $C(Q)$.

In this example, we tell Maxima to treat Q as a positive number by using `assume`, and we define the marginal cost MC in terms of a Maxima function (rather than a Maxima expression). We use the Maxima function `integrate` with a dummy variable of integration `qp`.

```
(%i4) killAB()$
      assume (Q > 0);
      MC(q) := 2*exp(0.2*q);
      Cf : 90;
      C : Cf + integrate (MC(qp), qp, 0, Q);

(%o1) [Q>0]
(%o2) MC(q) := 2 exp(0.2 q)
(Cf) 90
(C) 2 (5 %eQ/5 - 5) + 90

(%i5) C : expand (C);
(C) 10 %eQ/5 + 80
```

If we evaluate C for Q = 0 units, we get the fixed cost of production Cf = 90.

```
(%i6) at (C, Q = 0);
(%o6) 90
```

Of course if we then take the first derivative of C(Q) wrt Q, we should recover the marginal-cost function we started with.


```
(%i7) MC1 : diff (C, Q);
(MC1) 2 %eQ/5

(%i8) is (equal ( MC1, MC(Q) ));
(%o8) true
```

Here is a plot of C(Q) vs Q with a tangent line whose slope is the same as the local slope of the curve C(Q) at the point of tangency Q = 95.

```
(%i9) wxdraw2d (xlabel = "Q", ylabel = "C(Q)", key_pos = top_left, key = "C(Q)",
  explicit (C, Q, 85, 100), color = red, key = "tangent at Q = 95",
  explicit (MC(95)*(Q - 95) + subst (95, Q, C), Q, 92, 98), color = black, key = "",
  point_type = filled_circle, points ([95], [subst(95, Q, C)]) )$
```

(%t9)



4.6 Example: Average Cost vs Marginal Cost

Given a total-cost function $C(Q)$, the average-cost (AC) function is defined as

$$AC = C(Q)/Q, \text{ (for } Q > 0\text{),}$$

the ratio of total cost at a particular value of $Q > 0$, divided by that particular value of Q . This is the same as the average total cost ATC above.

Consider an example in Chiang & Wainwright, p. 160, in which $C(Q) = Q^3 - 12Q^2 + 60Q$, an expression with zero fixed cost.

```
(%i3) killAB()$
C : Q^3 - 12*Q^2 + 60*Q;
MC : diff (C,Q);
AC : expand (C/Q);
```

(C) $Q^3 - 12Q^2 + 60Q$

(MC) $3Q^2 - 24Q + 60$

(AC) $Q^2 - 12Q + 60$

The curves MC and AC cross at $Q = 6$.


```
(%i4) solve (MC = AC, Q);
```

```
(%o4) [Q=0, Q=6]
```

Because the expression for AC is finite at $Q = 0$, we can draw a plot over the range $Q = [0, 10]$.

```
(%i5) wxdraw2d (xlabel = "Q", ylabel = "MC, AC", xrange = [0,10], yrange = [0, 100],
  key_pos = top_left, key = "MC", explicit (MC, Q, 0, 10),
  color = red, key = "AC", explicit (AC, Q, 0, 10) )$
```

(%t5)



We proved the general theorem above which states that

- (a) the slope of the AC curve will be positive if and only if the MC curve lies above the AC curve,
- (b) the slope of the AC curve will be negative if and only if the MC curve lies below the AC curve,
- (c) the slope of the AC curve will be zero only at the point of intersection with the MC curve.

This theorem does NOT state that, when AC is negatively sloped, MC must also be negatively sloped; it merely says that AC must exceed MC in that circumstance.

In our example, to the left of $Q = 6$, AC is declining, and thus MC lies below it. To the right of $Q = 6$, AC is increasing and thus MC must lie above it.

4.7 Dowling 14.27

Find TC, AC, and VC if $MC = 25 + 30Q - 9Q^2$ and the fixed cost is 55.

```
(%i6) killAB()$
      assume (Q > 0);
      MC(q) := 25 + 30*q - 9*q^2;
      FC : 55;
      TC : FC + integrate (MC(qp), qp, 0, Q);
      AC : expand (TC/Q);
      VC : TC - FC;
```

```
(%o1) [Q>0]
```

```
(%o2) MC(q) := 25 + 30 q + (-9) q^2
```

```
(FC) 55
```


```
(TC) -3 Q^3 + 15 Q^2 + 25 Q + 55
```

```
(AC) -3 Q^2 + 15 Q +  $\frac{55}{Q}$  + 25
```

```
(VC) -3 Q^3 + 15 Q^2 + 25 Q
```

```
(%i7) wxdraw2d (explicit (TC,Q,0,50))$
```

```
(%t7)
```



4.8 Dowling 14.34

The marginal cost MC is $12 \exp(Q/2)$ and the fixed cost FC is 36. Find the total cost function TC(Q). Let cc be the constant of integration.

```
(%i3) killAB()$
      FC : 36;
      MC : 12*exp (Q/2);
      TC : cc + integrate (MC, Q);

(FC) 36
(MC) 12 %eQ/2
(TC) cc+24 %eQ/2

(%i4) soln : solve (subst (Q = 0, TC) = FC, cc);
(soln) [cc=12]

(%i5) TC : at (TC, soln);
(TC) 24 %eQ/2 + 12
```

Note that the constant of integration cc , in this example, does not equal the fixed cost FC .

4.9 Dowling 14.35, modified.

Let $MC = 16.3 \exp(0.48 Q)$ and $FC = 78.4$. Find $TC(Q)$.

```
(%i3) killAB()$
      FC : 78.4;
      MC : 16.3*exp (0.48*Q);
      TC : cc + integrate (MC, Q);

(FC) 78.4
(MC) 16.3 %e0.48 Q
(TC) cc+33.958 %e0.48 Q

(%i4) soln : solve (subst (Q = 0, TC) = FC, cc), numer;
(soln) [cc=44.442]

(%i5) TC : at (TC, soln);
(TC) 33.958 %e0.48 Q + 44.442
```

5 Total and Marginal Revenue

5.1 Find Marginal Revenue from a Demand Curve [Hands, Ch. 1]

Consider this example from Hands, Ch. 1, p. 19.

Assume a linear demand curve $Q = -P + 5$, where P is the market price per unit of good and Q is the quantity of units of that good. If $Q = 1$ unit means one thousand Toyota cars, then P is the market price per one thousand Toyota cars.

A linear demand curve is not "economics-realistic" as $Q \rightarrow 0$.

The "inverse demand curve" means P as a function of Q , $P = 5 - Q$.

The total revenue associated with each level of output Q is simply the price per unit (P) times the number of units (Q):

$$TR = TR(Q) = P Q = (5 - Q) Q = 5 Q - Q^2$$

The marginal Revenue (MR) is the first derivative of the total revenue with respect to Q .

$$MR = d(TR)/dQ = dTR/dQ = 5 - 2 Q.$$

Let MR_0 be the value of Q for which $MR = 0$.

An economist would label the first plot "Demand Curve" since economists have the habit of breaking the normal convention of (dependent variable=yaxis, independent variable = xaxis) in the case of a plot involving prices as the independent variable.


```
(%i8) killAB()$
TR : 5*Q - Q^2;
MR : diff (TR,Q);
MR0 : solve (MR, Q);
TRmax : at (TR, MR0);
MR0 : rhs (MR0[1]);
MRmin : at (MR, Q = 5);
wxdraw2d (xlabel = "Q", ylabel = "P", title = "Inverse Demand Curve",
  key = "P = 5 - Q", explicit (5 - Q, Q, 0, 5))$
wxdraw2d (xlabel = "Q", ylabel = "TR, MR", title = "Total Revenue and Marginal Revenue",
  yrange = [MRmin, 7], key_pos = bottom_left, key = "TR = 5*Q - Q^2",
  color = red, explicit (TR, Q, 0,5), color = purple,
  key = "MR = 5 - 2*Q", explicit (MR, Q, 0, 5), color = black, key = "",
  line_width = 1, explicit (0, Q, 0, 5), parametric (MR0, yy, yy, MRmin, TRmax) )$
```

(TR) $5Q - Q^2$

(MR) $5 - 2Q$


(MR0) $[Q = \frac{5}{2}]$

(TRmax) $\frac{25}{4}$


(MR0) $\frac{5}{2}$

(MRmin) -5

(%t7)



(%t8)



```
(%i9) float (TRmax);
(%o9) 6.25
```

When the price per unit is 4, the quantity demanded is 1 unit and the total revenue TR is 4. At the lower price $P = 1$ the quantity demanded is 4 units and the total revenue TR is again 4. We expect, from human nature, that the lower the price per unit, the more units will be sold.

Between the low Q and high price region and the high Q but low price region there is a "sweet spot" for maximum revenue, which occurs at the value of Q where MR passes from positive to negative.

5.2 Dowling 14.29, TR & P(Q) from MR

Given $MR = 60 - 2Q - 2Q^2$, find the total revenue TR and the inverse demand function $P(Q)$.

Since when $Q = 0$, $TR = 0$ also (by definition), we just need to integrate MR from $Q = 0$ to some arbitrary positive Q to find $TR(Q)$. We then can use $TR(Q) = P(Q)Q$ to solve for $P(Q)$, the inverse demand function. It is then clear that the average revenue function $AR(Q) = TR(Q)/Q$ is the same as the inverse demand function $P(Q)$.

```
(%i4) killAB()$
MR(q) := 60 - 2*q - 2*q^2;
assume (Q > 0);
TR : integrate (MR(q), q, 0, Q), expand;
P : expand (TR/Q);
```

```
(%o1) MR(q) := 60 - 2q + (-2)q^2
```

```
(%o2) [Q > 0]
```

```
(TR) -  $\frac{2Q^3}{3} - Q^2 + 60Q$ 
```

```
(P) -  $\frac{2Q^2}{3} - Q + 60$ 
```

For what value of $Q > 0$ is TR equal to its maximum?

```
(%i5) solns : solve (MR(Q), Q);
(solns) [Q = -6, Q = 5]
```


The only physical solution is $Q = 5 > 0$.

```
(%i6) soln : rest (solns, 1);  
(soln) [Q=5]  
  
(%i7) MR0 : 5;  
(MR0) 5  
  
(%i8) TRmax : at (TR, soln), numer;  
(TRmax) 191.67  
  
(%i9) Pmax : at (P, soln), numer;  
(Pmax) 38.333  
  
(%i10) Pmin : at (P, Q = 10), numer;  
(Pmin) -16.667  
  
(%i11) MRmin : at (MR(Q), Q = 10);  
(MRmin) -160  
  
(%i12) solns : solve (P, Q), numer;  
(solns) [Q=-10.266, Q=8.7664]  
  
(%i13) soln : rest (solns,1);  
(soln) [Q=8.7664]  
  
(%i14) QP0 : at (Q, soln);  
(QP0) 8.7664
```


```
(%i16) wxdraw2d (xlabel = "Q", ylabel = "P", title = "Inverse Demand Curve P(Q)",
    key = "P = 60 - Q - 2*Q^2/3", explicit (P, Q, 0, 10), color = black, key = "",
    line_width = 1, explicit (0, Q, 0, 10), explicit (Pmax, Q, 0, 5),
    parametric (5, yy, yy, Pmin, Pmax), parametric (QP0, yy, yy, Pmin, 0))$
```

```
wxdraw2d (xlabel = "Q", ylabel = "TR, MR", title = "Total Revenue and Marginal Revenue",
    yrange = [MRmin, 200], key_pos = bottom_left, key = "TR = 60*Q - Q^2 - 2*Q^3/3",
    color = red, explicit (TR, Q, 0, 10), color = purple,
    key = "MR = 60 - 2*Q - 2*Q^2", explicit (MR, Q, 0, 10), color = black, key = "",
    line_width = 1, explicit (0, Q, 0, 10), parametric (MR0, yy, yy, MRmin, TRmax),
    parametric (QP0, yy, yy, MRmin, 0))$
```

(%t15)



(%t16)



The total revenue reaches its maximum value just as the marginal revenue passes from positive to negative (the slope of the TR(Q) curve is zero at that point). The total revenue continues to increase as long as the marginal revenue remains > 0.

6 Consumption Function, Marginal Propensity to Consume

In Dowling, Ch.2, there is a discussion of national income Y determination models, such as $Y = C + I + G + (X - Z)$, with C = consumption, I = investment, G = government spending, X = exports, Z = imports.

Any one of the terms on the right hand side can have a component which is a function of national income Y , and which will hence influence the equilibrium value of Y .

In particular, for example, we can have $C = C_0 + b Y$, where C_0 is independent of national income Y , and b is the marginal propensity to consume MPC, the first derivative of consumption with respect to Y . However, b itself may be a function of Y .

6.1 Dowling 14.32, Find Consumption Function from MPC and C_0

Find the consumption function $C(Y) = C_0 + b g(Y)$ if $C_0 = 45$, and if $MPC = 0.6 + 0.1/Y^{1/3}$ = first derivative of $C(Y)$ wrt Y . We use an indefinite integral (anti-derivative) and add a constant c which can be adjusted to match the conditions at the point $Y = 0$.

```
(%i2) killAB()$
MPC : 0.6 + 0.1/Y^(1/3);
C : c + integrate (MPC, Y);
```

```
(MPC)  $\frac{0.1}{Y^{1/3}} + 0.6$ 
```

```
(C)  $c + 0.6 Y + 0.15 Y^{2/3}$ 
```

The first derivative of the above should equal MPC. Maxima treats c as a constant unless told otherwise using depends.

```
(%i3) diff (C, Y);
```

```
(%o3)  $\frac{0.1}{Y^{1/3}} + 0.6$ 
```

```
(%i4) soln : solve ( subst (Y=0, C) = 45, c);
```

```
(soln) [c=45]
```

```
(%i5) C : at (C, soln);
```

```
(C)  $0.6 Y + 0.15 Y^{2/3} + 45$ 
```

Thus $C(Y) = 45 + 0.6 Y + 0.15 Y^{2/3}$.

7 Savings Function and Marginal Propensity to Save

Reference: Hands, Ch. 3, pp 96-97.

In a closed economy (no foreign trade) the aggregate expenditure on goods is

$$C + I + G$$

where C = (personal) consumption expenditure, I = investment expenditure, G = government spending.

The aggregate income generated by the production of these goods is

$$C + S + T$$

where S = savings and T = taxes.

For the goods market to be in equilibrium, expenditure = income, so

$$C + I + G = C + S + T$$

or

$$S = I + (G - T).$$

National savings = investment - deficit, where

D = (G - T) = government spending - tax receipts = deficit.

In the IS-LM model, the savings function S depends on both the level of national income (GDP) Y and the rate of interest r: $S = S(Y, r)$, with the standard economic assumptions that $\partial S / \partial Y > 0$ and $\partial S / \partial r > 0$.

7.1 Dowling 14.33 Marginal Propensity to Save MPS and S(Y)

Suppose the marginal propensity to save MPS is

$$\partial S / \partial Y = 0.5 - 0.2/Y^{(1/2)} \text{ and when } Y = 25, S = -3.5.$$

Find S(Y).

```
(%i2) killAB()$
```

```
MPS : 0.5 - 0.2/Y^(1/2);
```

```
S : c + integrate (MPS, Y);
```

```
(MPS) 0.5 - \frac{0.2}{\sqrt{Y}}
```

```
(S) c + 0.5 Y - 0.4 \sqrt{Y}
```

```
(%i3) soln : solve (subst (Y = 25, S) = -3.5, c);
```

```
(soln) [c = -14]
```

(%i4) S : at (S, soln);

(S) $0.5 Y - 0.4 \sqrt{Y} - 14$

(%i5) at (S, Y = 25);

(%o5) -3.5

Thus the savings function is

$$S(Y) = 0.5 Y - 0.4 Y^{1/2} - 14.$$